

Lesson 15: Strange attractor

```
> restart;
with(plots):
N1:= proc(X)
local X1, t1, t5, t7, t12, t16, t22, t24;
X1:= X[1];
t1 := cos(X[2]);
t5 := X[2]^2;
t7 := X[1]^2;
t12 := 1/(2.*t1*X[1]*X[2]+t1*t5-t7-2.*X[1]*X[2]);
t16 := t7*X[2]+X[1]*t5-2.;
t22 := sin(X[2]);
t24 := t12*(X[1]+t22-2.);
X[1]:= X[1]-t1*t12*t16+X[1]*(X[1]+2.*X[2])*t24;
X[2]:= X[2]+t12*t16-X[2]*(2.*X1+X[2])*t24;
end proc:
NC:= Compiler[Compile](N1);
```

```
NC := proc( )
```

```
option call_external, define_external(_m0c941d081c25653ec83d0b5a865edebe, MAPLE,
LIB
```

```
= "C:\Users\daddy\AppData\Local\Temp\daddy-784\_m0c941d081c25653ec83d0b5a865ed\
ebeU1yYq2xV.dll");
```

```
call_external(0, 67113936, true, false, args)
```

```
end proc
```

(1)

A strange attractor

Looking around at what happens with different starting points, I found large regions that were attracted to one or the other of our two solutions, but also other regions where more complicated things happened. Here's one. I took this starting point:

```
> W[0]:= Vector(<1.0, -0.1>, datatype=float[8]);
```

$$W_0 := \begin{bmatrix} 1. \\ -0.1000000000000000 \end{bmatrix}$$

(1.1)

Now I want to use NC, my compiled version of Newt. Remember, this operates by changing its input. In order to do this, and still save the old value, I'll have to operate on a copy of the value.

Now something like this won't quite work:

```
> W[1]:= W[0]; NC(W[1]);
```

$$W_1 := \begin{bmatrix} 1. \\ -0.1000000000000000 \\ 2.22441889652819969 \end{bmatrix} \quad (1.2)$$

```
> W[0], W[1];
```

$$\begin{bmatrix} -0.212973067249683 \\ 2.22441889652820 \end{bmatrix}, \begin{bmatrix} -0.212973067249683 \\ 2.22441889652820 \end{bmatrix} \quad (1.3)$$

The problem was that assigning Vectors works a bit differently from ordinary expressions. What was assigned to W[1] was the Vector W[0], not just the value of W[0]. So when NC changed W[1], that also affected W[0]. Instead, we can use **copy** which will produce a copy of our Vector.

```
> W[0]:= Vector(<1.0, -0.1>, datatype=float[8]):
W[1]:= copy(W[0]):
NC(W[1]):
W[0], W[1];
```

$$\begin{bmatrix} 1. \\ -0.1000000000000000 \end{bmatrix}, \begin{bmatrix} -0.212973067249683 \\ 2.22441889652820 \end{bmatrix} \quad (1.4)$$

```
> for count from 1 to 20 do
  W[count]:= copy(W[count-1]);
  NC(W[count]);
end do;
> W[16],W[17],W[18],W[19],W[20];
```

$$\begin{bmatrix} 0.0111792772804370 \\ 13.0263547349452 \end{bmatrix}, \begin{bmatrix} 0.00881590251858578 \\ 14.7530817414898 \end{bmatrix}, \begin{bmatrix} 0.0116059228097792 \\ 12.7241087716031 \end{bmatrix},$$


$$\begin{bmatrix} 0.00895814170689274 \\ 14.5811203756396 \end{bmatrix}, \begin{bmatrix} 0.0125003478816499 \\ 12.0563097882685 \end{bmatrix} \quad (1.5)$$

The last few iterations look like they might be starting to settle down to a 2-cycle, but it doesn't happen. I'll give the process some more time to settle down, then save and plot the next 40000 points. I don't need to save the first 1000 points.

```
> X:= copy(W[20]):
for count from 1 to 1000 do
  NC(X)
end do;
Now to plot the next 50000 points, using pointplot from the plots package. You can give pointplot a Matrix whose columns are the points, with datatype=float[8]. Note how I can assign a column of the Matrix to be the Vector X (I don't need to use copy here).
```


```
> M:= Matrix(2,50000, datatype=float[8]):
for count from 1 to 50000 do
  NC(X);
  M[1..2,count]:= X;
end do;
```


```
> pointplot(M, symbol=point); PP:= %:
```




The points appear to lie on two curves that are folded over on themselves. On closer examination, it turns out that there seem to be an infinite number of "folds": this is an example of a so-called "strange attractor".


```
> display(PP,view=[.0085 .. .0095, 14.5 .. 15]);  
display([PP,plottools[rectangle]( [.0085,14.5],[.0095,15],  
colour=red)]);
```






```
> display(PP,view=[.00898 .. .00901, 14.56 .. 14.59]);  
display([PP,plottools[rectangle]( [.00898,14.56],[.00901,  
14.59],colour=red)],  
view=[.0085 .. .0095, 14.5 .. 15]);
```






For a closer look at what happens here, I'll follow 1001 initial points near the end of one of the curves (say on the line $[.010, 13.06]$ to $[.01, 13.07]$), and see what happens to them as we iterate our Newton function. This is using a different version of animation: here you produce each frame of the animation separately, and then give **display** a list of the frames, with the option **insequence=true**. Each of the 1001 points will have a slightly different colour, going through the rainbow from red to violet. Those can be specified as `COLOUR(HUE,c)` where `c` goes from 0 to 1 (but 1 is the same as 0)

```
> M:= Matrix(2,1001, datatype=float[8]):
  for ii from 1 to 1001 do
    M[1,ii]:= 0.01 ;
    M[2,ii]:= 13.06 + 0.00001*(ii-1);
  end do:
> colours:= evalf([seq(COLOUR(HUE,i/1100),i=0..1000)]):
frame[0]:= pointplot(M, symbol=point, colour=colours):
frame[0];
```



Now for each frame we apply NC to all the points, and plot the new positions of all the points.

```
> for count from 1 to 60 do
  for ii from 1 to 1001 do
    X:= M[1..2,ii];
    NC(X);
    M[1..2,ii]:= X;
  end do;
  frame[count]:= pointplot(M, symbol=point,colour=colours);
end do;
> display([seq(frame[count],count=0..60)], insequence=true);
```

The points start out very close together, but spread out to fill half of the attractor (and alternate between the two halves). One of the hallmarks of **chaos** is **sensitive dependence on initial conditions**: a very small difference in the initial point produces large differences after many iterations.

Integration

```
> restart;
```

The main integration command in Maple is **int**. It can be used for both antiderivatives and definite integrals:

```
> int(x^2, x);
```

$$\frac{1}{3} x^3$$

```
> int(x^2, x = 1 .. 4);
```

$$21$$

Iterated integrals can be done:

```
> int(int(x^2*y, x = 0 .. y), y = 0 .. 1);
```

$$\frac{1}{15}$$

There is also an inert integration command **Int**, for when you want to write down an integral formula but don't want to actually do the integration.

```
> Int(x^2, x = 1 .. 4);
```

$$\int_1^4 x^2 dx$$

A formula containing **Int** can be manipulated in various ways. When you want to actually do the integration, you can use the **value** command on any expression containing the **Int**.

```
> % = value(%);
```

$$\int_1^4 x^2 dx = 21$$

Antiderivatives

As you know, it can be hard to find an antiderivative. Maple is pretty good at it, but not perfect. Here's a hard one it can do.

```
> F := int(sqrt(tan(x)), x);
```

$$F := \frac{1}{2} \frac{\sqrt{\tan(x)} \cos(x) \sqrt{2} \arccos(\cos(x) - \sin(x)) - \frac{1}{2} \sqrt{2} \ln(\cos(x))}{\sqrt{\cos(x) \sin(x)}} + \sqrt{2} \sqrt{\tan(x)} \cos(x) + \sin(x)$$

```
> simplify(F);
```

$$\frac{1}{2} \frac{1}{\sqrt{\cos(x) \sin(x)}} \left(\sqrt{2} \left(\sqrt{\frac{\sin(x)}{\cos(x)}} \cos(x) \arccos(\cos(x) - \sin(x)) - \ln(\cos(x) + \sqrt{2} \sqrt{\frac{\sin(x)}{\cos(x)}} \cos(x) + \sin(x)) \sqrt{\cos(x) \sin(x)} \right) \right) \quad (3.1)$$

Should we believe this? The way to check an antiderivative is to differentiate it.

```
> dF := diff(F, x);
```

$$dF := \frac{1}{4} \frac{\cos(x) \sqrt{2} \arccos(\cos(x) - \sin(x)) (1 + \tan(x)^2)}{\sqrt{\tan(x)} \sqrt{\cos(x) \sin(x)}} - \frac{1}{4} \frac{\sqrt{\tan(x)} \cos(x) \sqrt{2} \arccos(\cos(x) - \sin(x)) (-\sin(x)^2 + \cos(x)^2)}{(\cos(x) \sin(x))^{3/2}} - \frac{1}{2} \frac{\sqrt{\tan(x)} \sin(x) \sqrt{2} \arccos(\cos(x) - \sin(x))}{\sqrt{\cos(x) \sin(x)}} - \frac{1}{2} \frac{\sqrt{\tan(x)} \cos(x) \sqrt{2} (-\sin(x) - \cos(x))}{\sqrt{\cos(x) \sin(x)} \sqrt{1 - (\cos(x) - \sin(x))^2}} - \frac{1}{2} \frac{1}{\cos(x) + \sqrt{2} \sqrt{\tan(x)} \cos(x) + \sin(x)} \left(\sqrt{2} \left(-\sin(x) \right. \right.$$

$$+ \frac{1}{2} \frac{\sqrt{2} \cos(x) (1 + \tan(x)^2)}{\sqrt{\tan(x)}} - \sqrt{2} \sqrt{\tan(x)} \sin(x) + \cos(x) \Bigg) \Bigg)$$

It doesn't look much like $\sqrt{\tan(x)}$, but maybe it simplifies.

> simplify(dF);


$$\begin{aligned} & -\frac{1}{2} \left(-2 \sin(x)^2 \sqrt{\cos(x) \sin(x)} + 2 \sin(x)^2 \cos(x)^2 \sqrt{\cos(x) \sin(x)} \right. \\ & - 2 \cos(x) \sin(x) \sqrt{\cos(x) \sin(x)} + 2 \cos(x)^3 \sin(x) \sqrt{\cos(x) \sin(x)} \\ & - \sqrt{\frac{\sin(x)}{\cos(x)}} \cos(x) \arccos(\cos(x) - \sin(x)) + \cos(x)^3 \arccos(\cos(x) - \sin(x)) \\ & - \sin(x) \sqrt{\frac{\sin(x)}{\cos(x)}} + \sin(x)^2 \cos(x) \sqrt{\frac{\sin(x)}{\cos(x)}} \arccos(\cos(x) - \sin(x)) \\ & - \sqrt{2} \cos(x) \sin(x) \sqrt{\cos(x) \sin(x)} \sqrt{\frac{\sin(x)}{\cos(x)}} \\ & + \sqrt{2} \cos(x)^3 \sin(x) \sqrt{\cos(x) \sin(x)} \sqrt{\frac{\sin(x)}{\cos(x)}} \\ & \left. - \sin(x)^3 \cos(x) \sqrt{\frac{\sin(x)}{\cos(x)}} \sqrt{\cos(x) \sin(x)} \sqrt{2} \right) / \\ & \left(\cos(x) \sin(x)^2 \sqrt{\cos(x) \sin(x)} \left(\sqrt{\frac{\sin(x)}{\cos(x)}} \cos(x) + \sqrt{2} \sin(x) \right. \right. \\ & \left. \left. + \sin(x) \sqrt{\frac{\sin(x)}{\cos(x)}} \right) \right) \end{aligned}$$

To check if $A = B$, rather than hoping Maple can simplify A to B , a better idea might be to try to simplify $A - B$ to 0 (or $\frac{A}{B}$ to 1).

> simplify(dF - sqrt(tan(x)));
0

This won't always work: checking whether a complicated expression is 0 is not an easy task. Another possibility is to plot the difference.

> plot(dF - sqrt(tan(x)), x = 0 .. Pi/2 - 0.0001, axes = box);



The differences are small: presumably due to roundoff error.

This one might be harder:

```
> J := Int(sqrt(sec(x)^2 + 4), x);
```

$$J := \int \sqrt{\sec(x)^2 + 4} \, dx$$

```
> value(J);
```

$$\left(\left(-\frac{2}{125} + \frac{1}{125} I \right) \sqrt{10} \left(10 \operatorname{EllipticF} \left(\frac{\left(\frac{2}{5} + \frac{1}{5} I \right) (-1 + \cos(x)) \sqrt{5}}{\sin(x)}, \frac{3}{5} - \frac{4}{5} I \right) \right. \right.$$

$$\left. \left. + 5 I \operatorname{EllipticF} \left(\frac{\left(\frac{2}{5} + \frac{1}{5} I \right) (-1 + \cos(x)) \sqrt{5}}{\sin(x)}, \frac{3}{5} - \frac{4}{5} I \right) \right)$$

$$\begin{aligned}
& -4 \operatorname{EllipticPi} \left(\frac{\sqrt{\frac{3}{5} + \frac{4}{5} I} (-1 + \cos(x))}{\sin(x)}, \frac{3}{5} - \frac{4}{5} I, \frac{\sqrt{\frac{3}{5} - \frac{4}{5} I}}{\sqrt{\frac{3}{5} + \frac{4}{5} I}} \right) \\
& -2 I \operatorname{EllipticPi} \left(\frac{\sqrt{\frac{3}{5} + \frac{4}{5} I} (-1 + \cos(x))}{\sin(x)}, \frac{3}{5} - \frac{4}{5} I, \frac{\sqrt{\frac{3}{5} - \frac{4}{5} I}}{\sqrt{\frac{3}{5} + \frac{4}{5} I}} \right) \\
& -16 \operatorname{EllipticPi} \left(\frac{\sqrt{\frac{3}{5} + \frac{4}{5} I} (-1 + \cos(x))}{\sin(x)}, -\frac{3}{5} + \frac{4}{5} I, \frac{\sqrt{\frac{3}{5} - \frac{4}{5} I}}{\sqrt{\frac{3}{5} + \frac{4}{5} I}} \right) \\
& -8 I \operatorname{EllipticPi} \left(\frac{\sqrt{\frac{3}{5} + \frac{4}{5} I} (-1 + \cos(x))}{\sin(x)}, -\frac{3}{5} + \frac{4}{5} I, \frac{\sqrt{\frac{3}{5} - \frac{4}{5} I}}{\sqrt{\frac{3}{5} + \frac{4}{5} I}} \right) \\
& \cos(x) \sin(x)^2 \sqrt{-\frac{-40 \cos(x) + 20 I \cos(x) - 10 - 20 I}{1 + \cos(x)}}
\end{aligned}$$

$$\left(\sqrt{\frac{4 \cos(x) + 2 I \cos(x) + 1 - 2 I}{1 + \cos(x)}} \sqrt{\frac{1 + 4 \cos(x)^2}{\cos(x)^2}} \right) / \left(\sqrt{\frac{3}{5} + \frac{4}{5} I} (4 \cos(x)^3 - 4 \cos(x)^2 + \cos(x) - 1) \right)$$

It's not just complicated, but it involves some special functions (*EllipticPi* and *EllipticF*). Is it correct?

```
> simplify(diff(value(J),x)-sqrt(sec(x)^2 + 4));
```

0

(3.2)

But actually there's a relatively simple, elementary solution that Maple misses (I'm not sure exactly why).

There is a **Student[Calculus1]** package that can help with integration at the Math 101 level. You can, for example, ask it for a hint on how to do an integral.

```
> with(Student[Calculus1]):
```

```
> Hint(J);
```

[change, u = tan(x), u]

It's telling us to try the change of variables $u = \tan(x)$. It will also do that for you.

```
> J1:= Rule[%](J);
```

$$J1 := \int \sqrt{\sec(x)^2 + 4} \, dx = \int \frac{\sqrt{5 + u^2}}{u^2 + 1} \, du$$

What next?

> **Hint(%);**

$$[change, u1 = \sqrt{5 + u^2} - u, u1], [change, u = \sqrt{5} \tan(u1), u1]$$

One of these looks interesting, but I'd prefer a different letter s instead of the name $u1$ Maple chose for the new variable.

> **J2:=Rule[change,s=sqrt(5+u^2)-u,s](J1);**

$$J2 := \int \sqrt{\sec(x)^2 + 4} \, dx = \int \frac{-10s^2 - 25 - s^4}{25s - 6s^3 + s^5} \, ds$$

OK, that's a rational function. We might tackle that with partial fractions. But let's just let Maple do it.

> **value(rhs(%));**

$$2 \arctan(2 + s) - \ln(s) - 2 \arctan(-2 + s)$$

Now we have to substitute back to get the integral in terms of u , and then the original variable x .

> **eval(%,s = sqrt(u^2+5)-u);**

$$-\ln(\sqrt{5 + u^2} - u) - 2 \arctan(-2 - \sqrt{5 + u^2} + u) + 2 \arctan(-\sqrt{5 + u^2} + u + 2)$$

> **G:= eval(%,u = tan(x));**

$$G := -\ln(\sqrt{5 + \tan(x)^2} - \tan(x)) + 2 \arctan(2 + \sqrt{5 + \tan(x)^2} - \tan(x)) - 2 \arctan(\sqrt{5 + \tan(x)^2} - \tan(x) - 2)$$

That's our answer. Let's check it.

> **dG:= diff(G,x);**

$$dG := -\frac{\frac{\tan(x) (1 + \tan(x)^2)}{\sqrt{5 + \tan(x)^2}} - 1 - \tan(x)^2}{\sqrt{5 + \tan(x)^2} - \tan(x)} + \frac{2 \left(\frac{\tan(x) (1 + \tan(x)^2)}{\sqrt{5 + \tan(x)^2}} - 1 - \tan(x)^2 \right)}{1 + (2 + \sqrt{5 + \tan(x)^2} - \tan(x))^2} - \frac{2 \left(\frac{\tan(x) (1 + \tan(x)^2)}{\sqrt{5 + \tan(x)^2}} - 1 - \tan(x)^2 \right)}{1 + (\sqrt{5 + \tan(x)^2} - \tan(x) - 2)^2}$$

> **simplify(dG - sqrt(sec(x)^2+4));**
0

Some antiderivatives just can't be expressed in "closed form"

> **int(exp(sin(x)), x);**

$$\int e^{\sin(x)} dx$$

Others need special functions. This one is important in probability.

```
> int(exp(-x^2/2), x);
```

$$\frac{1}{2} \sqrt{\pi} \sqrt{2} \operatorname{erf}\left(\frac{1}{2} \sqrt{2} x\right)$$

Maple objects introduced in this lesson

copy

insequence=true (option for **display** to produce an animation)

COLOUR(HUE,...) (option for colour in plot commands)

int

Int

value

indets

lprint

Student[Calculus1] package

Hint (in **Student[Calculus1]**)

Rule (in **Student[Calculus1]**)

change (in **Student[Calculus1]**)